
Abstract. In this article, I discuss the need for Halton sequences and discuss
the Mata implementation of scrambling of Halton sequences, providing several
examples of scrambling procedures.

Keywords: st0244, Halton sequence, quasi–Monte Carlo, scrambled Halton se-
quence

1 Halton sequences

Many areas of scientific simulation rely on methods for generating random or pseudoran-
dom numbers from the interval [0, 1) or the unit cube [0, 1)s in s dimensions. A common
application is multivariate integration. Suppose an integral I(f) =

∫
[0,1)s f(u) du needs

to be evaluated. Integration over different domains can be handled by appropriate trans-
formations as is done in the Geweke–Hajivassiliou–Keane estimator (see [M-5] ghk( )).
In the context of random variables, integration may be handled by the transformation
from the range of the random variables into quantiles that are marginally distributed
as U [0, 1]. A sequence ui (i = 1, . . . , N ; 0 ≤ uil < 1; l = 1, . . . , s) can be used to
approximate the integral I(f) as

Î =
1
N

N∑
i=1

f(ui) (1)

provided that the numbers u1,u2, . . . behave in a way similar to uniform numbers. In
fact, multivariate integration does not require the draws to be independent, and a regu-
lar distribution of the finite sequence over [0, 1)s plays a more important role. Although
the usual pseudorandom numbers, such as those generated by Stata’s runiform()
random-number generator, will serve the purpose, somewhat more accurate results can
be achieved with quasi–Monte Carlo methods (Niederreiter 1992; Lemieux 2009).

Let us compare the two panels in figure 1. The plot on the left uses the pseudorandom
numbers generated by runiform(). As is natural for random data, some points are
clumped together and some areas are left relatively unpopulated. In contrast, the plot
on the right, generated using the Halton sequence (described in the next paragraph),
does not show such artifacts and provides a highly uniform coverage of the space. From
the point of view of numeric integration, if the points are clumped together, they will
likely have close values of the evaluated functions, adding little additional information
to (1). Computing time would be better spent spreading points farther apart. At the
same time, in the areas that are left sparsely populated, some important features of

c© 2012 StataCorp LP st0244



30 Scrambled Halton sequences

the integrand may be missed. Thus the Halton sequence will likely work better in
applications that require a uniform spread of points but not their independence.

set obs 50
set seed 10203
generate rx = runiform()
generate ry = runiform()
scatter ry rx, aspect(1)
graph export Halton1a.eps, replace
generate qmcx = .
generate qmcy = .
mata: st_store(., ("qmcx qmcy"), halton(50, 2))
scatter qmcy qmcx, aspect(1)
graph export Halton1b.eps, replace

Figure 1. Random and quasi–Monte Carlo points in [0, 1)2

To define the Halton sequence, we need to make a short excursion into number
theory and the presentation of numbers in different bases. For an integer b ≥ 2, positive
integers can be represented in base b using the relation

n = a0(n) + a1(n)b + a2(n)b2 + · · · + ak(n)bk

=
k∑

j=0

aj(n)bj , 0 ≤ aj(n) < b for all j = 0, . . . , k

which can be represented graphically as (akak−1 . . . a1a0)b. The upper bar indicates that
this is a sequence of digits in base b rather than a product of numbers (its dependence
on n suppressed to reduce clutter). A radical inverse function in base b is a reflection
of this representation with respect to the decimal point: (0.a0a1 . . . ak−1ak)b, that is,

φb(n) =
k∑

j=0

aj(n)b−j−1

Note that φb(n) ∈ [0, 1). For an integer b ≥ 2, the van der Corput sequence in base b is
the sequence {φb(n)}∞n=0. A generalization of van der Corput sequences into multiple

dimensions is provided by the Halton sequence (Halton 1960). For a dimension s, let
b1, . . . , bs be integers greater than 1. Then the Halton sequence in the bases b1, . . . , bs

is the sequence
xn = {φb1(n), . . . , φbs

(n)}

The quantitative description of the regularity properties of sequences such as pseu-
dorandom numbers or Halton sequences is provided by various forms of discrepancies.
They measure the difference between the frequency with which the sequence passes
through a certain set and the Lebesgue measure (multidimensional volume) λ(·) of this
set. Let the number of hits of set B by sequence P be denoted by

A(B,P ) =
N∑

n=1

1B(xn)

where 1B(x) is an indicator function of a set B, 1B(x) = 1 if x ∈ B, and 1B(x) = 0
otherwise. For a collection of sets B, define the discrepancy as the greatest difference
between the Lebesgue measure and the proportion of hits over all sets in B:

DN (B;P ) = sup
B∈B

∣∣∣A(B,P )
N − λ(B)

∣∣∣
The family of s-dimensional boxes I∗ = {[0, v1) × · · · × [0, vs) : v1, . . . , vs < 1} starting
from the origin and the family I = {[u1, v1) × · · · × [us, vs) : 0 ≤ ul < vl < 1 for all l =
1, . . . , s} of arbitrary s-dimensional boxes in [0, 1)s are two leading examples:

D∗
N (P ) = DN (I∗;P ), I∗ =

{
s∏

l=1

[0, ul) : 0 ≤ ul ≤ 1

}

DN (P ) = DN (I;P ), I =

{
s∏

l=1

[ul, vl) : 0 ≤ ul ≤ vl ≤ 1

}

D∗
N (P ) is usually referred to as the star discrepancy and DN (P ) as the extreme dis-

crepancy. The relation between them is D∗
N (P ) ≤ DN (P ) ≤ 2sD∗

N (P ).

It has been established (Niederreiter 1992) that the asymptotic discrepancy of the
van der Corput sequence in base b is

lim
N→∞

ND∗
N (Sb)

lnN = lim
N→∞

NDN (Sb)
lnN =

⎧⎪⎪⎨⎪⎪⎩
b2

4(b + 1) ln b
, b ∈ 2N

b − 1
4 ln b

, b ∈ 2N + 1



32 Scrambled Halton sequences

Assuming that S is the Halton sequence in pairwise relatively prime bases b1, . . . , bs,
then (Atanassov 2004)

D∗
N (S) <

s

N +
1
N

s∏
l=1

(
bl − 1
2l ln bl

lnN +
bl + 1

2

)
= A(b1, . . . , bs)N−1 lns N + O

(
N−1 lns−1 N

)
A(b1, . . . , bs) =

1
s!2s

s∏
l=1

bl − 1
ln bl

→ 0 as s → ∞ (2)

This discrepancy features a better asymptotic rate O(N−1 lns N ) than the random
Monte Carlo rate O(N−1/2) that arises from the central limit theorem.

Implementation of the Halton sequence is available in Mata (Drukker and Gates
2006) with the halton() suite of functions. These functions allow for the generation of
van der Corput sequences with an arbitrary base, as well as the generation of Halton
sequences skipping the initial entries. See [M-5] halton( ).

The Halton sequence is one of the examples of low-discrepancy sequences studied
in the area of quasi–Monte Carlo sampling. Other sequences that provide the same
asymptotic discrepancy rate are known. Among these, the Halton sequence is the
only one in which an extra point uN+1 can be added quite naturally. With other
sequences, the points need to be added in large blocks to improve the actual finite-
length discrepancy.

2 Scrambled Halton sequences

Each of the margins of the Halton sequence, that is, the van der Corput sequence
in base bl, is organized in blocks of consecutive points at the distance 1/bl from each
other, somewhere in the corresponding intervals [0, 1/bl), [1/bl, 2/bl), . . . . This behavior
generates an undesirable artifact observed in short sequences with larger dimensions.

Consider figure 2. The left panel reproduces the scatterplot of the first 50 elements
of Halton sequences in the first two dimensions corresponding to b1 = 2 and b2 = 3.
It shows the regularity that we are seeking. The right panel gives the scatterplot of
the first 50 elements in dimensions 7 and 8 corresponding to the primes b7 = 17 and
b8 = 19. Contrary to the plausible expectations of regularity, the points form a very
strong pattern on parallel lines with slope 17/19 = 0.895. Is it possible to avoid this
problem while retaining the desirable regularity properties of the Halton sequence?

clear
set obs 50
foreach p of numlist 2 3 5 7 11 13 17 19 {
generate h`p´ = .
local hlist `hlist´ h`p´
}
mata: st_store(., tokens("`hlist´"), halton(50, 8))
scatter h3 h2, aspect(1)
graph export Halton2a.eps, replace
scatter h19 h17, aspect(1)
graph export Halton2b.eps, replace

Figure 2. Initial elements of the Halton sequences in [0, 1)2

The answer is yes. The asymptotic discrepancy properties of Halton sequences are
retained when an extra scrambling operation is performed when the radical inverse is
taken. Namely, let σl : {0, 1, . . . , bl−1} �→ {0, 1, . . . , bl−1} be a permutation of numbers
0, 1, . . . , bl − 1. If the radical inverse is taken of [σl{a0(n)}, σl{a1(n)}, . . . , σl{ak(n)}]b
instead of (a0a1 . . . ak)b, it produces a number

φb,σl
(n) =

∞∑
j=0

σl{aj(n)}b−j−1

The scrambled Halton sequence, with the given set of permutations σ1(·), . . . , σs(·), is
then obtained by collecting the scrambled sequences over margins.

What is a “good” permutation, and how can one be found? Certain permutations
were shown to improve the multiplicative constant A in front of the leading term in
asymptotic discrepancy (2). They are, however, very cumbersome to derive and code.
Let us instead motivate several simple options.

One of the reasons that prime numbers are chosen as the bases for Halton sequences
is that for any prime p, the residues 0, 1, . . . , p − 1 form an Abelian group with respect
to the operation (x + y) mod p (Dummit and Foote 2004). Any nonzero element can
act as the generator of the group, meaning that if x is a nonzero element, then the set of



34 Scrambled Halton sequences

numbers {0, x mod p, 2x mod p, . . . , (p − 1)x mod p} coincides with {0, 1, . . . , p − 1}.
Hence, a simple permutation rule is provided by

σ(a) = xa mod p, x �= 0, a ∈ {0, 1, . . . , p − 1} (3)

Let us introduce the necessary Mata tools, and then give some examples.

3 Syntax

The Mata library lscrhalton.mlib distributed with this article contains the following
functions:

• Mata functions to generate prime numbers:

– PrimesUpTo(n) returns all prime numbers less than or equal to n.

– FirstPrimes(n) returns the first n primes.

• Mata functions to convert bases:

– tolongbase(b, n) returns a real vector containing the b-digits of representa-
tion of n in base b.

– fromlongbase(b, v) returns a real value corresponding to the vector v.

– The functions tolongbase() and fromlongbase() are mutually reciprocal.

• Mata functions to generate scrambled Halton sequences:

– ScrHalton(n, d, &f(), optional arguments) generates a scrambled Hal-
ton sequence of length n in d dimensions. Scrambling is performed by the
user-written function f(), whose syntax must conform to specifications given
below.

– ScrHalton(k, b, &f(), optional arguments) returns a scalar between 0
and 1 obtained by expanding the integer k in base b, applying the scrambling
procedure defined in the user-written function f(), and computing the radical
inverse of the result. Up to five additional parameters are supported and
passed through to f().

The user-written function f() must satisfy the following requirements. Its syntax is
f(aj, b, j, optional arguments), where aj is the integer number between 0 and b− 1
to be scrambled, b is the base of the transformation, j is the position of the digit in
the representation of an underlying number in base b, and optional arguments can be
passed through to f(). It will return the scrambled number between 0 and b − 1, and
it will leave the value 0 intact.

4 Examples

Examples given in this section operate by replacing the call to Stata’s standard halton()
function in the code snippet from page 33. Thus we shall give the body of the scram-
bler function and an example of the call. The complete code is provided in the do-file
scrambled-halton.do distributed with the article. The data preparation commands
and the plotting commands are the same as in the aforementioned code. When repro-
ducing these examples, make sure that all the dimensions of the resulting scrambled
Halton sequences match those of the data. In this example, there are 50 observations
on 8 variables.

Example 1: Square-root scrambler

As the first example, let us use 	
√

bj
 as the generator of the set {1, . . . , bj − 1},
that is, as the multiplier x in (3), with p = bj . Here 	·
 is Stata’s floor() function,
which computes the nearest smaller integer of the argument.

real scalar SqrtScrambler(real scalar k, real scalar b,
real scalar j) {

return(mod(k*floor(sqrt(b)), b))
}
st_store(., tokens("h2 h3 h5 h7 h9 h11 h13 h17 h19"),

ScrHalton(50,8,&SqrtScrambler())

The resulting plot of the selected pairs of dimensions is given in figure 3. Both
graphs show reasonably regular behavior, although the graph of the higher dimensions
with greater bases still demonstrates the points on parallel lines.

Figure 3. Initial elements of the Halton sequences in [0, 1)2 scrambled with the square-
root multiplier

Example 2: Negative square-root scrambler

An equally reasonable option is to use
{
−
[√

bj

]}
mod bj = bj −

[√
bj

]
as the

generator. As a slight modification, the nearest integer to the square root rather than
the nearest smaller integer is used in the code below:

real scalar MinusSqrtScrambler(real scalar k, real scalar b,
real scalar j) {

return(mod(k*(b-round(sqrt(b),1)), b))
}
st_store(., tokens("h2 h3 h5 h7 h9 h11 h13 h17 h19"),

ScrHalton(50,8,&MinusSqrtScrambler())

The resulting plot of the selected pairs of dimensions is given in figure 4, with results
similar to those of figure 3. Arguably, the performance of the scrambled sequence for
the higher two primes is improved, although some regular patterns remain.

Figure 4. Initial elements of the Halton sequences in [0, 1)2 scrambled with the negative
square-root multiplier

Example 3: Random multiplier scrambler

To demonstrate the flexibility of the code in supplying additional options, let us
consider the following weak version of randomization: draw the multiplier x in (3)
uniformly from {1, 2, . . . , bj − 1}. To ensure that the same permutation is applied
consistently to all elements of the scrambled Halton sequence, the random seed is reset
each time the procedure is called. This seed acts as a parameter of the procedure and
is supplied to ScrHalton().

real scalar RandomMultipleScrambler (real scalar k, real scalar b,
real scalar j, real scalar seed) {

real scalar mult;

rseed(seed)
mult = ceil(runiform(1,1)*(b-1))
return(mod(k*mult, b))

}
st_store(., tokens("h2 h3 h5 h7 h9 h11 h13 h17 h19"),

ScrHalton(50,8,&RandomMultipleScrambler(),10203)

The resulting plot of the selected pairs of dimensions is given in figure 5. The reader
is encouraged to try different values of the seed and observe the resulting differences.

Figure 5. Initial elements of the Halton sequences in [0, 1)2 scrambled with the random
multiplier

Example 4: Random permutation scrambler

To make more extensive use of the randomization possibilities, we can generate the
whole permutation σl(·) randomly. To ensure that the same permutation is applied
consistently to all elements of the scrambled Halton sequence, the random seed is reset
each time the procedure is called. This seed acts as a parameter of the procedure and
is supplied to ScrHalton().

real scalar RandomPermuteScrambler(real scalar k, real scalar b,
real scalar j, real scalar seed) {

real colvector permut

rseed(seed)
permut = (0, jumble(1::(b-1)))
return(permut[k+1])

}
st_store(., tokens("h2 h3 h5 h7 h9 h11 h13 h17 h19"),

ScrHalton(50,8,&RandomPermuteScrambler(),1234576)

The resulting plot of the selected pairs of dimensions is given in figure 6. Arguably,
the scatterplots are more characteristic of the pseudorandom number sequences like
those in figure 1(a), with clumping of points and empty regions of the plot. Although
the useful regularity properties of the Halton sequence will be preserved asymptotically,
this property is not guaranteed for short series.

Figure 6. Initial elements of the Halton sequences in [0, 1)2 scrambled with the random
multiplier

Example 5: Atanassov’s modified Halton sequence

An advanced example of scrambling procedures based on the abstract algebra con-
cepts comes from Atanassov (2004). He proposed a permutation that depends on the
position of the digit after the decimal point as

σlj(a) = akj
l mod bj (4)

for specially chosen numbers kl related to the primitive roots in the field Zbj
. The set of

“good” numbers is hard-coded in the routine, and is transferred back and forth between
the scrambler and the ScrHalton() routine as a Mata matrix.

real scalar AtanassovGHaltonScrambler(real scalar k, real scalar b,
real scalar j, real matrix K) {

real colvector sel, kj, prod, i;

if (K==.) {
K = (2, 1 \ 3, 1 \ 5, 4 \ 7, 2 \ 11, 9 \ 13, 9 \ 17, 2 \ 19, 1 \ 23, 13)
K = (K \ 29, 6 \ 31, 22 \ 37, 7 \ 41, 37 \ 43, 36 \ 47, 36 \ 53, 39 \ 59, 4)
K = (K \ 61, 26 \ 67, 13 \ 71, 12 \ 73, 35 \ 79, 66 \ 83, 60 \ 89, 68)
K = (K \ 97, 63 \ 101, 47 \ 103, 15 \ 107, 104 \ 109, 4 \ 113, 64)

}

sel = !(K[.,1] :- b)
kj = select(K, sel)[1,2]

prod = 1
for(i=1; i<=j; i++) {

prod = mod(kj*prod, b)
}

return(mod(k*prod,b))
}
st_store(., tokens("h2 h3 h5 h7 h9 h11 h13 h17 h19"),

ScrHalton(50,8,&AtanassovGHaltonScrambler(),K=.)

The resulting plot of the selected pairs of dimensions is given in figure 7. Although
the strong lined patterns of figure 2(b) disappeared, other patterns are present in the
higher dimensions.

Figure 7. Initial elements of the Halton sequences in [0, 1)2 scrambled with the
Atanassov multiplier

Example 6: Braaten and Weller’s permutations

The first impetus to work with scrambled Halton sequences came from an article
by Braaten and Weller (1979). To find good permutations, they used a greedy search
algorithm that minimizes another version of discrepancy, the L2 discrepancy T ∗

N . The
latter can be computed explicitly.

This scrambler consists of three functions. The main function called by ScrHalton()
is BraatenWellerScrambler(); it uses an optional argument, Mata matrix M, to store
the “good” permutations. These permutations are found by BWPermutation(), which
in turn calls Tstar() to compute the discrepancy of a sequence. An optional starting
value of the permutation can be provided to BWPermutation(), which must be a number
from [0, 1). The default value of 0 reproduces the results of Vandewoestyne and Cools
(2006), who reproduced most of the sequences from Braaten and Weller (1979) or pro-
duced other sequences with identical discrepancies.

real scalar BraatenWellerScrambler(real scalar k, real scalar b,
real scalar j, real matrix M, | real scalar start) {

// M has permutation stored by rows;
// the first column is the associated prime
real rowvector perm;
real scalar whereb;

// starting point
if (start == .) start = 0;
else if (start<0 | start>=1) return(.);

if (max(!(M[.,1]:-b))==0) {
// initialize the permutation for the prime b
perm = BWPermutation(b,start)
if (M==.) M = (b, perm)
else M = (M, J(rows(M),cols(perm)-cols(M)+1,.) (b, perm))

}
else {

// find where the permutation is stored
whereb = sum((M[.,1]:<=b))
perm = M[whereb,2..b+1]

}
return(perm[k+1])

}

real rowvector BWPermutation(real scalar b, | real scalar start) {

real rowvector perm;
real scalar i, j, argmin, minT, currentT;

// starting point
if (start == .) start = 0;
else if (start<0 | start>=1) return(.);

perm = floor(start*b)
// ensure that 0 maps to 0
if (perm != 0) perm = (0, perm)

for(i=1;i<b;i++) {
// add a point to minimize the overall discrepancy
// of the sequence accumulated so far
minT = .
for(j=0;j<b;j++) {

// is j in the perm vector already?
if (sum(!(perm:-j))) continue;
else {

// compute discrepancy if the point j is added
currentT = _Tstar((perm:/b,j:/b))
if (currentT < minT) {

argmin = j
minT = currentT

}
}

}
perm = (perm, argmin)

}
return((0, perm))

}
// compute the T-star discrepancy of a series
real scalar _Tstar (real rowvector x) {

real scalar i, j, sum1, sum2, N;

N = cols(x)
sum1 = sum2 = 0
for(i=1;i<=N;i++) {

if(x[i]<0 | x[i]>=1) return(.)
for(j=1;j<=N;j++) {

sum1 = sum1 + 1-max((x[i], x[j]))
}
sum2 = sum2 + (1-x[i]*x[i])

}
return(sum1/(N*N) - sum2/N + 1/3)

}

st_store(., tokens("h2 h3 h5 h7 h9 h11 h13 h17 h19"),
ScrHalton(50,8,&BraatenWellerScrambler(),M=.)

st_store(., tokens("h2 h3 h5 h7 h9 h11 h13 h17 h19"),
ScrHalton(50,8,&BraatenWellerScrambler(),M=.,0.5)

As a by-product of this procedure, the Mata matrix M contains the permutations
found by Braaten–Weller’s algorithm. The resulting plot of the selected pairs of dimen-
sions is given in figure 8. The top row corresponds to the default starting value of 0,
and the bottom row corresponds to the custom starting value of 0.5. The plots in higher
dimensions are not ideal because some areas are sparsely populated.

0
.2

.4
.6

.8
1

D
im

en
si

on
 2

, b
=3

0 .2 .4 .6 .8 1
Dimension 1, b=2

(a) Dimensions 1 and 2, b1 = 2 and b2 = 3;

(b) Dimensions 7 and 8, b7 = 17 and b8 = 19;

(c) Dimensions 1 and 2, b1 = 2 and b2 = 3;
starting point is �b/2�

(d) Dimensions 7 and 8, b7 = 17 and b8 = 19;

Figure 8. Initial elements of the Halton sequences in [0, 1)2 scrambled with the Braaten–
Weller procedure

5 Discussion

Although scrambling Halton sequences, as described above, helps overcome “autocorre-
lations” of the original Halton sequences with higher values of the prime bases, scram-
bling is not a panacea: some artifacts may remain or new artifacts may be introduced.
For reviews of different scrambling procedures, see Vandewoestyne and Cools (2006),
Schlier (2008), and Faure and Lemieux (2009).

In some applications where Halton sequences or their generalizations are used to
approximate multivariate integrals, a measure of accuracy of the approximation is re-
quired. It can be obtained by randomizing the sequence so that each point is dis-
tributed uniformly on [0, 1)s, and taking several sequences with independent random
starting points. If the sequence u(k)

i , i = 1, . . . , N , is produced from a random start-
ing point k = 1, . . . , K and the integral I(f) =

∫
[0,1)s f(u) du is approximated by

Îk = 1/N
∑N

i=1 f(u(k)
i ), then the overall approximation to the integral in question is

Î = 1/K
∑

k Îk, and its variance is estimated by s2
I = 1/(K − 1)

∑
k(Îk − Î)2.

A simple randomization rule is to add a uniform random number (mod 1) to all
elements of the (scrambled) Halton sequence. More advanced procedures will “continue”
the Halton sequence from a random starting point by using an alternative representation
of the iterations between the consecutive elements of the Halton sequence using additions
with carry-over in base b. Of course, either idea can be combined with scrambling.

